多品種輸送問題を解く

- 多品種輸送問題の最適化（例1）
  - 3つの工場で4種の製品A,B,C,Dを作っている．ただし，工場1はBとD，工場2はA,B,C，工場3はB,C,D のみを生産できる
  - 各工場の生産可能量は製品の種類に関係なく，それぞれ最大3000個
  - 5人の顧客に必要量を輸送する．輸送コスト最小となる輸送計画をたてたい

＜各工場で生産可能な製品と合計生産可能量＞

工場＼製品	*A*	*B*	*C*	*D*	生産可能量
1	―	*OK*	―	*OK*	*3000*
2	*OK*	*OK*	*OK*	―	*3000*
3	―	*OK*	*OK*	*OK*	*3000*

＜製品1単位あたり輸送コストと需要＞

	工場輸送コスト			製品需要			
顧客	*1*	*2*	*3*	*A*	*B*	*C*	*D*
α	*4*	*6*	*9*	*80*	*85*	*300*	*6*
β	*5*	*4*	*7*	*270*	*160*	*400*	*7*
γ	*6*	*3*	*4*	*250*	*130*	*350*	*4*
δ	*8*	*5*	*3*	*160*	*60*	*200*	*3*
ε	*10*	*8*	*4*	*180*	*40*	*150*	*5*

- 最適化問題の定式化（変数/係数）
  - 変数x_{ijk}：工場j→顧客i の製品k輸送量
  - 生産可能量の係数ベクトル $\boldsymbol{b}$
  - 輸送コストを表す係数行列 $\boldsymbol{C}$
  - 需要を表す係数行列 $\boldsymbol{D}$

多品種輸送問題を解く

- 最適化問題の定式化（ベタ表記）

$$\text{min. } 4(x_{111}+x_{112}+x_{113}+x_{114}) + \ldots + 4(x_{531}+x_{532}+x_{533}+x_{534})$$

$$\text{s. t. } x_{111}+x_{121}+x_{131}=80,\ x_{112}+x_{122}+x_{132}=85$$

$$x_{113}+x_{123}+x_{133}=300,\ x_{114}+x_{124}+x_{134}=6$$

$$\ldots$$

$$(x_{111}+x_{112}+x_{113}+x_{114})+\ldots+(x_{511}+x_{512}+x_{513}+x_{514}) \leqq 3000$$

$$(x_{121}+x_{122}+x_{123}+x_{124})+\ldots+(x_{521}+x_{522}+x_{523}+x_{524}) \leqq 3000$$

$$(x_{131}+x_{132}+x_{133}+x_{134})+\ldots+(x_{531}+x_{532}+x_{533}+x_{534}) \leqq 3000$$

$$\left.\begin{array}{l} (x_{111}+x_{113})+\ldots+(x_{511}+x_{513})=0 \\ (x_{124})+\ldots+(x_{524})=0 \\ (x_{131})+\ldots+(x_{531})=0 \end{array}\right\} \text{使わない変数を0に（各工場で生産してない製品を0に）}$$

$$x_{111},\ldots,x_{534} \geqq 0$$

- 最適化問題の定式化（Σ表記）

$$\text{min.} \sum_{i=1}^{5}\sum_{j=1}^{3}\sum_{k=1}^{4} c_{ij}x_{ijk}$$

$$\text{s. t. } \sum_{j=1}^{3} x_{ijk} = d_{ik} (i = 1, \ldots, 5; k = 1, \ldots, 4)$$

$$\sum_{i=1}^{5}\sum_{k=1}^{4} x_{ijk} \leq b_j (j = 1, \ldots, 3)$$

$$\sum_{i=1}^{5}\sum_{k=1}^{4} m_{jk}x_{ijk} = 0 (j = 1, \ldots, 3)$$

$$x_{ijk} \geq 0 \ (i = 1, \ldots, 5; j = 1, \ldots, 3; k = 1, \ldots, 4)$$

$$\boldsymbol{b} = (b_1 \quad b_2 \quad b_3) = (3000 \quad 3000 \quad 3000)$$

$$\mathrm{C} = \begin{pmatrix} c_{11} & c_{12} & c_{13} \\ c_{21} & c_{22} & c_{23} \\ c_{31} & c_{32} & c_{33} \\ c_{41} & c_{42} & c_{43} \\ c_{51} & c_{52} & c_{53} \end{pmatrix} = \begin{pmatrix} 4 & 6 & 9 \\ 5 & 4 & 7 \\ 6 & 3 & 4 \\ 8 & 5 & 3 \\ 10 & 8 & 4 \end{pmatrix}$$

$$\mathrm{D} = \begin{pmatrix} d_{11} & d_{12} & d_{13} & d_{14} \\ d_{21} & d_{22} & d_{23} & d_{24} \\ d_{31} & d_{32} & d_{33} & d_{34} \\ d_{41} & d_{42} & d_{43} & d_{44} \\ d_{51} & d_{52} & d_{53} & d_{54} \end{pmatrix} = \begin{pmatrix} 80 & 85 & 300 & 6 \\ 270 & 160 & 400 & 7 \\ 250 & 130 & 350 & 4 \\ 160 & 60 & 200 & 3 \\ 180 & 40 & 150 & 5 \end{pmatrix}$$

$$\mathrm{M} = \begin{pmatrix} m_{11} & m_{12} & m_{13} & m_{14} \\ m_{21} & m_{22} & m_{23} & m_{24} \\ m_{31} & m_{32} & m_{33} & m_{34} \end{pmatrix} = \begin{pmatrix} 1 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 \end{pmatrix}$$

How to use CPLEX?

- 新規プロジェクトの作成
  ① [ファイル(F)]－[新規(N)]－[OPLプロジェクト]を選択
  ② [プロジェクト名] を記入（例：MultiTransport）し，3カ所にチェックする
    ☑ デフォルトの実行構成の追加
    ☑ モデルの作成
    ☑ データの作成
  ③ [終了]をクリック

プロジェクト名は自由だが，半角英数で何の問題を解こうとしているのかが分かる名前が良い

- プロジェクト内のいくつかの名前を変更
  - ✓ [構成1] → [config1]　※日本語を英語に変更しないと実行時エラーになる
  - ✓ モデルファイル [MultiTransport.mod] → [mt.mod]
  - ✓ データファイル [MultiTransport.dat] → [mtex1.dat]

- モデルファイル・データファイルを記述し保存（次ページ参照）
- [config1]にモデルファイルとデータファイルをセットする

How to use CPLEX?

- mt.mod

```
int i_max = ...;// 顧客数
int j_max = ...;// 工場数
int k_max = ...;// 製品数

range I = 1..i_max;// 顧客集合の範囲
range J = 1..j_max;// 工場集合の範囲
range K = 1..k_max;// 製品集合の範囲

int c[I,J] = ...; // 工場 j → 顧客 i への製品1単位あたり輸送コスト
int d[I,K] = ...; // 顧客 i の 製品 k の需要量
int b[J] = ...;   // 工場 j の生産可能合計量
int m[J,K] = ...; // 工場 j の 製品 k の生産有無を表す係数行列

dvar float+ x[I,J,K];// 工場 j → 顧客 i への 製品 k の輸送量(非負)

minimize
  sum(i in I) sum(j in J) sum(k in K) c[i,j]*x[i,j,k];
subject to {
  forall(i in I) {
    forall(k in K) {
      sum(j in J) x[i,j,k] == d[i,k];
    }
  }
  forall(j in J) {
    sum(i in I) sum(k in K) x[i,j,k] <= b[j];
    sum(i in I) sum(k in K) m[j,k]*x[i,j,k] == 0;
 }
}
```

How to use CPLEX?

- mtex1.dat

```
i_max = 5;
j_max = 3;
k_max = 4;

b = [3000 3000 3000];
c = [
[4 6 9]
[5 4 7]
[6 3 4]
[8 5 3]
[10 8 4]
];
d = [
[ 80  85 300 6]
[270 160 400 7]
[250 130 350 4]
[160  60 200 3]
[180  40 150 5]
];
m = [
[1 0 1 0]
[0 0 0 1]
[1 0 0 0]
];
```

How to use CPLEX?

工場＼製品	*A*	*B*	*C*	*D*	生産可能量
1	—	*OK*	—	*OK*	*3000*
2	*OK*	*OK*	*OK*	—	*3000*
3	—	*OK*	*OK*	*OK*	*3000*

- 計算結果の確認([解]タブ)

最適値

```
// solution (optimal) with objective 12014
// Quality There are no bound infeasibilities.
// There are no reduced-cost infeasibilities.
// Maximum Ax-b residual              = 0
// Maximum c-B'pi residual            = 0
// Maximum |x|                        = 400
// Maximum |slack|                    = 2902
// Maximum |pi|                       = 8
// Maximum |red-cost|                 = 8
// Condition number of unscaled basis = 1.4e+02
x = [[[  0  85   0   6]      …工場1(B=85,D=6)
      [ 80   0 300   0]      …工場2(A=80,C=300)
      [  0   0   0   0]]     …工場3(0)
     [[  0   0   0   7]
      [270 160 400   0]
      [  0   0   0   0]]
     [[  0   0   0   0]
      [250 130 350   0]
      [  0   0   0   4]]
     [[  0   0   0   0]
      [160   0   0   0]
      [  0  60 200   3]]
     [[  0   0   0   0]
      [180   0   0   0]
      [  0  40 150   5]]];
```

顧客αへの輸送

Optimal solution 最適解

	製品需要			
顧客	*A*	*B*	*C*	*D*
α	*80*	*85*	*300*	*6*
β	*270*	*160*	*400*	*7*
γ	*250*	*130*	*350*	*4*
δ	*160*	*60*	*200*	*3*
ε	*180*	*40*	*150*	*5*

多品種輸送問題を解く

- 多品種輸送問題の最適化（例2） ※データはExcelファイル
  - ○つの工場で△種の製品を作っている．ただし，各工場で生産できる製品の種類は異なる
  - 各工場の生産可能量は製品の種類に関係なく，それぞれ最大◇個
  - □人の顧客に必要量を輸送する．輸送コスト最小となる輸送計画をたてよ

- 線形最適化問題として定式化し，CPLEXで解く
  - モデルファイル[mt.mod]は共通なので，データファイル[mtex2.dat]を作り解く